Rezolvarea numerică a ecuațiilor algebrice și transcendente

Vartic Cătălina, clasa a 12-a

Obiectivele proiectului:

- recunoaşte prezenţa soluţiilor unei ecuaţii algebrice sau transcendente pe un interval dat;
- separa intervalele domeniului de definiţie a unei funcţii f(x), care vor conţine exact o soluţie a ecuaţiei f(x) = 0;
- utiliza algoritmii de rezolvare a ecuaţiilor algebrice şi transcendente prin metoda bisecţiei, metoda coardelor şi metoda tangentelor;
- elabora programe de rezolvare a ecuaţiilor algebrice şi transcendente prin metoda bisecţiei, metoda coardelor şi metoda Newton;

Generalități

Se consideră ecuația de forma f(x)=0, unde funcția f: I $\rightarrow$ R ,este continua pe intervalul I.

- Orice valoare ξ, pentru care expresia f(ξ) = 0 este adevărată, se numeşte **zerou** al funcţiei f(x) sau **soluţie** a ecuaţiei f(x) = 0
- Ecuația f(x)=0 se numește **algebrică,** dacă pentru calcularea valorii funcției f(x), după valoarea dată x, se folosesc doar operații aritmetice și cea de ridicare la putere cu exponent rațional. Ecuația are nu este algebrică se numește ecuație **transcendentă**.

Exemple

Ecuații algebrice:

- $x^3 - 9*x^2 + 24*x - 19$;
- $x^5 - 5*x + 7$;
- $3x^2 - 18*x + 24$.

Ecuații transcendente:

- $\lg x + x^2 - \mathrm{tg}\, x + 2$;
- $2^x - 3*x + 1/x$;
- $\cos x + \sin x$.

Procesul de calcul numeric al rădăcinilor reale ale ecuațiilor de forma f(x)=0 se divizează în două etape principale:

1. ***Izolarea (separarea, localizarea) rădăcinilor***, adică determinarea unor intervale suficient de înguste, astfel încât fiecare din ele să conțină câte o singură rădăcină a ecuației f(x)=0

2. ***Micşorarea*** pe cît mai mult posibil a fiecărui din aceste intervale (dacă se pune problema determinării tuturor soluţiilor) sau a unuia din ele (dacă trebuie de determinat doar una din soluţii).

Izolarea rădăcinilor

- Deseori, la izolarea rădăcinilor se aplică ***metoda grafică.*** Pentru aceasta se construiește graficul funcției y=f(x) și, dacă ξ este abscisă puntului de intersecție a graficului cu axa **Ox, atunci f(**ξ)=0. De exemplu, în figura 3.2.1, abscisele punctelor de intersecție(ξ1, ξ2, ξ3) ale curbei y=f(x) cu axa Ox reprezintă rădăcinele ecuației f(x)=0



Fig. 3.2.1

Izolarea rădăcinilor

Fie există f’(x) pe I. Atunci izolarea rădăcinilor poate fi realizată prin ***metoda analitică***, care presupune următoarea succesiune de acțiuni:

1. Se află derivata f’(x);
2. Se rezolvă ecuația f’(x)=0;
3. Se compune tabelul semnelor valorilor f(x);
4. Se selectează intervalele, la extremitățile cărora funcția are valori de semne opuse. Acestea conțin cîte o singură rădăcină a ecuației în studiu.

Izolarea rădăcinilor

- **Exemplul 1** Să se izoleze grafic rădăcinile reale ale ecuației **f(x)=3*x^2-1=0**
- **Rezolvare.** Construim graficul funcției y=3*x^2-1. Conform figurii, ecuația are doua rădăcini reale ξ1, care aparține (-1,0) și ξ2, care aparține (0,1). Întrucît o ecuație de gradul doi posedă exact 2 rădăcini, iar în azul dat toate doua sunt reale, decade necesitatea demonstrării uniităţii lor pe intervalele respetive.



Izolarea rădăcinilor

- **Exemplul 2** Să se localizeze prin metoda analitiă rădăcinile reale ale ecuației **f(x)=x^3+3*x^2-1=0,** pe intervalul [-3,2].
- **Rezolvare.** Funcția f(x) este derivabilă pe intervalul [-3,2]. Găsim f'(x)=3*x^2+6*x. Rezolvăm ecuația f'(x)=0 : 3*x^2+6*x=0 ; x(x+2)=0; x1=0; x2=-2. Alcătuim tabelul semnelor funcției:

x	**-3**	**-2**	**-1**	**0**	**1**	**2**
f(x)	-	+	+	-	+	+

- Deoarece au loc 3 alternanțe de semne ale funcției f(x), conchidem că ecuația are 3 rădăcini reale în studiu, care se conțin respetiv în intervalele: (-3,-2); (-1,0); (0,1).

Metoda bisecției

Una dintre cele mai simple metode de determinare a unei soluţii a ecuaţiei f(x) = 0 este ***metoda bisecţiei***. Metoda presupune determinarea punctului de mijloc c al segmentului [a, b], apoi calculul valorii f(c). Dacă f(c) = 0, atunci c este soluţia exactă a ecuaţiei.

În caz contrar, soluţia este căutată pe unul dintre segmentele [a, c] şi [c, b]. Ea va aparţine

segmentului pentru care semnul funcţiei în extremităţi este diferit. Dacă f(a) × f(c) > 0, atunci soluţia e căutată în continuare pe segmentul [a1, b1], unde a1 primeşte valoarea c, iar b1 – valoarea b. În caz contrar, a1 primeşte valoarea a, iar b1 – valoarea c. Procesul de divizare se reia pe segmentul [a1, b1], repetîndu-se pînă cînd nu se obţine soluţia exactă sau devierea soluţiei calculate c_i de la cea exactă nu devine suficient de mică.



Eroarea metodei

$$\left|\xi - c_i\right| < \varepsilon = \left|b_i - a_i\right|.$$

Algoritmizarea metodei

Pornind de la descrierea matematică a metodei, putem separa două cazuri distincte de oprire a procesului de calcul al soluţiei ecuaţiei $f(x)=0$ pentru metoda bisecţiei:

A1. Algoritmul de calcul pentru un număr prestabilit n de divizări consecutive:

Pasul 0. Iniţializare: $i \Leftarrow 0$.

Pasul 1. Determinarea mijlocului segmentului $c \Leftarrow (a+b)/2$.

Pasul 2. Reducerea segmentului ce conţine soluţia: dacă $f(c) = 0$, atunci soluţia calculată este $x = c$. SFÎRŞIT.

În caz contrar, dacă $f(a) \times f(c) > 0$, atunci $a \Leftarrow c$; $b \Leftarrow b$, altfel $a \Leftarrow a$; $b \Leftarrow c$.

Pasul 3. $i \Leftarrow i + 1$. Dacă $i = n$, atunci soluţia calculată este $x=(a+b)/2$. SFÎRŞIT.

În caz contrar, se revine la pasul 1.

Algoritmizarea metodei

A2. Algoritmul de calcul pentru o precizie2 ε dată:

Pasul 1. Determinarea mijlocului segmentului $c \Leftarrow (a+b)/2$.

Pasul 2. Dacă $f(c) = 0$, atunci soluţia calculată este $x = c$. SFÎRŞIT.

În caz contrar, dacă $f(a) \times f(c) > 0$, atunci $a \Leftarrow c$; $b \Leftarrow b$, altfel $a \Leftarrow a$; $b \Leftarrow c$.

Pasul 3. Dacă $|b - a| < \varepsilon$, atunci soluţia calculată este $x=(a+b)/2$.

SFÎRŞIT.

În caz contrar, se revine la pasul 1.

Metoda bisecției

- **<u>Exemplul 1</u>** Să se determine o rădăcină a ecuaţiei $x4 + 2x3 - x - 1 = 0$ pe segmentul [0, 1] pentru 16 divizări consecutive.

```
program cn05;
var    a,b,c: real;
       i,n:integer;

function f(x:real):real;
begin  f:=sqr(sqr(x))+2*x*sqr(x)-x-1;end;
begin  a:=0; b:=1; n:=16;
       for i:=1 to n do
         begin c:=(b+a)/2;
            writeln('i=',i:3,' x=',c:10:8,' f(x)=',f(c):12:8);
            if f(c)=0 then break[3]
            else if f(c)*f(a)>0 then a:=c else b:=c;
         end;
end.
```

Rezultate

```
i= 1 x=0.50000000 f(x)= -1.18750000
i= 2 x=0.75000000 f(x)= -0.58984375
...
i= 15 x=0.86679077 f(x)= 0.00018565
i= 16 x=0.86677551 f(x)= 0.00009238
```

Metoda coardelor

Fie dată funcția f(x), care posedă următoarele proprietăți:

1. f(x) continuă pe segmentul [a,b] și f(a)*f(b)<0;
2. pe segmentul [a,b] există f'(x) diferit de zero și f''(x) diferit de zero.

Metoda coardelor presupune alegerea în calitate de aproximare a soluției punctului determinat de intersecția dreptei ce trece prin punctele (a,f(a)) și (b,f(b)) cu axa Ox.

Primul grafic - Apropierea succesivă de soluția ecuației. Extremitatea fixă – b.

Al doilea grafic - Apropierea succesivă de soluția ecuației. Extremitatea fixă – a.



Eroarea metodei

$$|\xi - x_i| \le \left|\frac{M_1 - m_1}{m_1}\right| \times |x_i - x_{i-1}|,$$

SAU

$$\left|\frac{M_1 - m_1}{m_1}\right| \times |x_i - x_{i-1}| \le \varepsilon.$$

ξ – soluţia exactă a ecuaţiei f(x) =0 pe segmentul [a, b],

M1 şi m1 – marginea superioară şi inferioară a f ′(x) pe acelaşi segment

Algoritmizarea metodei

A1. Algoritmul de calcul pentru un număr prestabilit n de aproximări succesive:

Pasul 1.Determinarea extremităţii fixe e şi a aproximării x_0 : $c \Leftarrow a - \dfrac{f(a)}{f(b)-f(a)}(b-a);$

dacă $f(c) \times f(a) < 0$, atunci $e \Leftarrow a$, $x_0 \Leftarrow b$, altfel $e \Leftarrow b$, $x_0 \Leftarrow a$; $i \Leftarrow 0$.

Pasul 2.Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \dfrac{f(x_i)}{f(e)-f(x_i)}(e-x_i).$

Pasul 3.Dacă $i + 1 = n$, atunci soluţia calculată $x \Leftarrow x_i$. SFÎRŞIT.

În caz contrar, $i \Leftarrow i+1$ şi se revine la pasul 2.

Algoritmizarea metodei

A2. Algoritmul de calcul pentru o exactitate ε dată:

Deoarece în formula de estimare a erorii figurează mărimile M1 şi m1 , atunci cînd valorile lor nu sînt indicate în enunţul problemei, este necesară descrierea analitică a f ′(x) şi calcularea M1 şi m1 .

Pasul 1. Determinarea extremităţii fixe e şi a aproximării x_0 :

$c \Leftarrow a - \frac{f(a)}{f(b)-f(a)}(b-a);$ dacă f(c) × f(a) < 0, atunci e ⇐ a, x_0 ⇐ b, altfel e ⇐ b, x_0 ⇐ a; i⇐0.

Pasul 2. Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f(e)-f(x_i)}(e-x_i).$

Pasul 3. Dacă $\left|\frac{M_1 - m_1}{m_1}\right| \times |x_{i+1} - x_i| \leq \varepsilon$, atunci soluţia calculată x ⇐ xi . SFÎRŞIT.

În caz contrar, i ⇐ i+1 şi se revine la pasul 2.

Metoda coardelor

<u>Exemplul 1</u> Fie dată funcţia f(x) = ln(xsinx). Să se calculeze soluţia aproximativă a ecuaţiei f(x) = 0 pe segmentul [0,5; 1,5] pentru 10 aproximări succesive, utilizînd metoda coardelor.

```
program cn07;
var   a,b,e,c,x: real;
      n,i: integer;

function f(x:real):real;
begin f:=ln(x*sin(x));end;
begin a:=0.5; b:=1.5; n:=10;
      {determinarea extremitatii fixe e si a aproximarii initiale x0}
      c:=a-(f(a))/(f(b)-f(a))*(b-a);
      if f(c)*f(a)>0 then begin e:=b; x:=a; end
      else begin e:=a; x:=b; end;
        {calculul iterativ al solutiei}
        for i:=1 to n do
           begin x:= x-(f(x))/(f(e)-f(x))*(e-x);
        writeln(x:10:8,' ',f(x):12:8);
           end;
end.
```

Rezultate

```
i= 1 x=1.27995775 f(x)= 0.20392348
i= 2 x=1.18251377 f(x)= 0.09028687
...
i= 9 x=1.11427651 f(x)= 0.00016577
i= 10 x=1.11420523 f(x)= 0.00006678
```

Metoda coardelor

Exemplul 2: Fie dată funcţia f(x) = x4 – 3x2 + 7,5x – 1. Să se calculeze soluţia aproximativă a ecuaţiei f(x) = 0 pe segmentul [–0,5; 0,5] cu exactitatea ε = 0,0001, utilizînd metoda coardelor. Pentru funcţia dată pe [–0,5; 0,5] M1 şi m1 sînt, respectiv, egale cu 10 şi 5.

```
program cn08;
var
   Msup,minf,a,b,e,x,xnou,xvechi,eps: real;
function f(x:real):real;
begin
   f:=sqr(sqr(x))-3*sqr(x)+7.5*x-1;
end;
begin
   a:=-0.5; b:=0.5; eps:=0.0001;
   Msup:=10; minf:=5;
   {determinarea extremitatii fixe si a aproximarii initiale}
   x:=a-(f(a))/(f(b)-f(a))*(b-a);
   if f(x)*f(a)>0 then begin e:=b; xnou:=a; end
   else begin e:=a; xnou:=b; end;
   {calculul iterativ al solutiei}
   repeat
   xvechi:=xnou;
   xnou:= xvechi-(f(xvechi))/(f(e)-f(xvechi))*(e-xvechi);
   writeln(' x=',xnou:10:8,' f(x)=',f(xnou):12:8);
   until abs((Msup-minf)/minf*(xnou-xvechi))<eps;
end.
```

Rezultate

```
x=0.22500000 f(x)= 0.53818789
x=0.15970438 f(x)= 0.12191694
...
x=0.14130134 f(x)= 0.00026052
x=0.14127062 f(x)= 0.00005579
```

Metoda Newton

Fie dată funcția f(x), care posedă următoarele proprietăți:

1. f(x) continuă pe segmentul [a,b] și f(a)*f(b)<0;
2. pe segmentul [a,b] există f’(x) diferit de zero și f’’(x) diferit de zero, continui, și semnul lor pe [a,b] este constant.



Se va încerca rezolvarea problemei prin trasarea consecutivă a unor tangente la graficul funcţiei, prima dintre ele fiind construită prin extremitatea E_0 (x_0 , y_0) a segmentului [a, b], extremitate pentru care se respectă condiţia: $f(x_0) \times f''(x_0) > 0$.

Fie că tangenta cu numărul i intersectează axa 0x în punctul xi. Următoarea tangentă (i+1) va fi trasată prin punctul E_{i+1} cu coordonatele (xi, f (xi)) şi va intersecta axa absciselor în punctul x_{i+1}. Şirul de valori x_0, x1, x2, ..., x_i, x_{i+1}, ... va converge către soluţia ecuaţiei f (x) = 0. Această metodă de calcul al soluţiei ecuaţiei f (x)=0 este numită metoda tangentelor sau Newton.

Eroarea metodei

$$\varepsilon = \left|\xi - x_{i+1}\right| \leq \frac{M_2}{2m_1}(x_{i+1} - x_i)^2$$

unde xi , xi+1 – două aproximări succesive ale soluţiei calculate,
M2 – supremul f ″(x) pe [a, b],
m1 – infimul f′(x) pe [a, b].

Algoritmizarea metodei

A1. Algoritmul de calcul pentru un număr dat de aproximări succesive:

Pentru a realiza acest algoritm, este suficient să fie cunoscute descrierile analitice pentru f (x) şi f ′(x). Dacă descrierea f ′(x) nu este indicată în enunţ, urmează să fie calculată. Aproximarea iniţială se deduce utilizînd procedeul similar determinării extremităţii fixe pentru metoda coardelor.

Pasul 1.Determinarea aproximării iniţiale x_0: $c \Leftarrow a - \frac{f(a)}{f(b)-f(a)}(b-a);$

dacă $f(c) \times f(a) < 0$, atunci $x_0 \Leftarrow a$, altfel $x_0 \Leftarrow b$: $i \Leftarrow 0$.

Pasul 2. Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$.

Pasul 3. Dacă $i+1 = n$, atunci soluţia calculată $x \Leftarrow x_{i+1}$. SFIRŞIT.

În caz contrar, $i \Leftarrow i+1$, apoi se revine la pasul 2.

Algoritmizarea metodei

A2. Algoritmul de calcul pentru o exactitate ε dată:

În formula de estimare a erorii figurează mărimile M2 şi m1. Atunci cînd valorile lor nu sînt indicate în enunţul problemei, este necesară o preprocesare matematică pentru stabilirea M2 şi m1. Suplimentar sînt necesare descrierile analitice pentru f (x) şi f ′(x).

Pasul 1. Determinarea aproximării iniţiale x_0: $c \Leftarrow a - \frac{f(a)}{f(b)-f(a)}(b-a);$

dacă f(c) × f(a) < 0, atunci $x_0 \Leftarrow a$, altfel $x_0 \Leftarrow b$: $i \Leftarrow 0$.

Pasul 2. Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$.

Pasul 3. Dacă $\frac{M_2}{2m_1}(x_{i+1}-x_i)^2 \le \varepsilon$, atunci soluţia calculată $x \Leftarrow x_{i+1}$. SFÎRŞIT.

În caz contrar, $i \Leftarrow i+1$ şi se revine la pasul 2.

Metoda Newton

<u>Exemplul 1</u> Fie dată funcţia f(x) = x3 – 2x2 + x – 3. Să se scrie un program care va calcula soluţia ecuaţiei f(x) = 0 pe segmentul [2, 15] pentru 10 aproximări succesive, utilizînd metoda Newton.

```
program cn09;
var a, b, x, c : real;
   i, n: integer;
function f(z:real):real;
      begin f:=z*z*z-2*z*z+z-3; end;
function fdl(z:real):real;
      begin fdl:=3*z*z-4*z+1; end;
begin     a:=2.1; b:=15; n:=10; i:=0;
          c:=a-(f(a))/(f(b)-f(a))*(b-a);
          if f(c)*f(a)<0 then x:=a else x:=b;
          while i<n do
          begin   i:=i+1;
                  x:=x-f(x)/fdl(x);
          writeln('i=',i:2,' x=',x:15:12, ' f=',f(x):15:12);
          end;
end.
```

Rezultate

```
i= 1 x= 10.23214285700  f=869.11072454000
i= 2 x= 7.06207637180   f=256.52261987000
...
i= 9 x= 2.17455942470   f= 0.00000009329
i=10 x= 2.17455941030   f= 0.00000000001
```

Metoda Newton

<u>Exemplul 2</u> Fie dată funcţia $f(x) = \cos^2(x) - \frac{x}{4}$ Să se scrie un program care va calcula soluţia aproximativă a ecuaţiei f(x) = 0 pe segmentul [2,4; 3] cu exactitatea ε = 0,0001, utilizînd metoda Newton. Pentru funcţia dată pe segmentul [2,4; 3] M2 şi m1 sînt, respectiv, egale cu 2 şi 0,03.

```
program cn10;
var a, b, xn, xv, M2, m1, e, c : real;
function f(z:real):real;
   begin f:=cos(z)*cos(z)-z/4;  end;
function fd1(z:real):real;
   begin fd1:=-sin(2*z)-1/4;  end;
begin     a:=2.4; b:=3; M2:=2; m1:=0.03; e:=0.0001;
          c:=a-(f(a))/(f(b)-f(a))*(b-a);
          if f(c)*f(a)<0 then begin
          xn:=a; xv:=b;
          end
          else begin xn:=b; xv:=a; end;
   while M2*sqr(xn-xv)/(2*m1)>e do
   begin   xv:=xn;
           xn:=xv-f(xv)/fd1(xv);
           writeln(' x=',xn:15:12, '  f=',f(xn):15:12);
   end;
end.
```

Rezultate

```
x= 2.47538619170 f= -0.00078052066
x= 2.47646766320 f= -0.00000027700
x= 2.47646804730 f= 0.00000000000
```

Bibliografie

- Botoșanu Mihail,Informatica manual pentru clasa a 12-a,Epigraf,2008
- Gremalschi Anatol,Informatică manual pentru clasa a 12-a,Știința,2015